Behavior Research Methods & Instrumentation
1982, Vol. 14(4), 422-424

PROGRAM ABSTRACTS/ALGORITHMS

An interactive FORTRAN IV program for calculating aspects of power with dichotomous data

WILLIAM P. DUNLAP
Tulane University, New Orleans, Louisiana 70118

Dunlap (1981) described an interactive program that calculates power, necessary sample size, or detectable differences among means with data appropriate for analysis of variance. The present paper describes a similar interactive program (see Appendix) for helping plan experiments with dichotomous data when the usual method of analysis is chi square. Rather than solving for power with contingency tables of any arbitrary dimensionality, this program focuses on contingency tables with two rows and two columns, where the null hypothesis can be expressed in terms of the difference between two population proportions. Although power calculations with higher order contingency tables are not particularly difficult (see Cohen, 1977), the present program was limited to 2 by 2 tables for several reasons. First, in higher order tables, specifying the anticipated pattern of frequencies from among the almost limitless number of possible patterns seemed a confusing task for the user. Why not target on two crucial groups with the simplest form of discrete data, dichotomous? Second, with the 2 by 2 table, exact power can be computed for the chi-square test, providing an index of accuracy for the various power-approximating algorithms. The question of accuracy is of no small concern for chi-square power calculations. Finally, three distinct power-approximating algorithms exist for the 2 by 2 problem; only one was found for higher order tables.

The chi-square test applied to frequency data is an approximate rather than exact test and involves three distinct steps of approximation in its derivation (see Fry, 1938). Because of this fact, an investigator can have a perfectly accurate power computation for the chi-square distribution, and still have poor accuracy in using the distribution in approximating probabilities associated with binomial data. Also, the commonly recommended Yate's correction for continuity is actually not appropriate for 2 by 2 contingency tables of the type ordinarily encountered by researchers, in that its use results in serious underestimates of Type I error rates and, thus, unnecessary loss of power (see Camilli & Hopkins, 1978). For this reason, power calculations with the program presented here would not be appropriate for the continuity-corrected 2 by 2 chi-square procedure. The first step in writing this program was to compare the various algorithms in terms of accuracy. To accomplish this, it was necessary to compute exact power values.

Exact Power Computations. To obtain exact power values for the 2 by 2 chi-square test, a program was written that generates all possible outcomes for an experiment with fixed column frequencies; row frequencies are free to vary as a function of the data sampled. This is the most commonly encountered 2 by 2 experimental design, labeled Case 2 by Pearson (1947).

As an example, compare a group of 10 females and 10 males on a binomial variable such as cigarette smoking (smokers vs. nonsmokers). For the males and for the females, there are 11 possible frequency patterns, ranging from all smokers to no smokers. Thus, a total of 121 possible patterns of outcome exist for the study, all of which are easily enumerated by the computer. Given theoretical population proportions for the groups, one can compute the probability of each group frequency pattern, and also, the joint probability for the particular table as the product of the group probabilities via the binomial distribution, as follows:

pattern probability =

$$\frac{n_1!}{s_1!(n_1 - s_1)!} P_1^{s_1} (1 - P_1)^{(n_1 - s_1)} \times$$

$$\frac{n_2!}{s_2!(n_2 - s_2)!} P_2^{s_2} (1 - P_2)^{(n_2 - s_2)},$$

where n is the sample size, P is the population proportion, s is the frequency of one category of the dichotomous data, and the subscripts refer to Groups 1 and 2 respectively. For each possible contingency table, the chi-square value can be computed, as well as the sum of the exact binomial probabilities of all tables whose chi-square values surpass the tabled critical values. Thus, the user computes the probability of obtaining tables that result in significance under the given population conditions, the power of the chi-square test. The only problem with using the exact power computation is that large sample sizes result in a prohibitive number of possible outcome patterns; thus, some approximating algorithm is needed.

Approximating Algorithms. Three approximating algorithms were tested for possible inclusion in the present program. The first, used by Cohen (1977) for power tables for 2 by 2 chi-square problems, uses the arcsin transformation to normalize the proportions and then calculates power from the normal curve. Comparison with actual power calculations shows this algorithm to be inferior in accuracy to two other procedures. The triangles in Figure 1 depict the accuracy of power computed with the approximation described by Zelen and Severo (1965, p. 942, Formula 26.4.28). The circles in Figure 1 show calculations for the same data

  0005-7878/82/040422-03$00.55/0



Figure 1. Average absolute deviation from exact power of two power-approximating algorithms (triangles represent Zelen & Severo, 1965; circles represent Fleiss, 1973) as functions of sample size and average population proportion, termed eccentricity.

using a formula from Fleiss (1973, p. 30, Formula 3.12). The ordinate of the figure represents accuracy as the absolute deviation of approximate power from exact power. Sample size per group is represented on the abscissa. The depth axis in Figure 1, labeled "eccentricity," is the average of the two population proportions studied at each point. Thus, if power were computed for Group 1 having a population proportion of .9 vs. Group 2 with a population proportion of .7, an eccentricity value of .8 would be plotted. Power is symmetric about .5 with respect to this parameter, so that population proportions of .1 and .3 result in identical power. Thus, "eccentricity" describes how far the average population proportion deviates from .5. It is obvious from Figure 1 that high values of eccentricity result in large amounts of error regardless of which algorithm is used. This is because the chi-square test for dichotomous data becomes conservative whenever the average population proportion is very large or very small, and this problem persists even for relatively large sample sizes. Therefore, at great eccentricity, the approximations rather badly overestimate power, and yet, no alternative algorithm compensates for this bias. The user of the program is, therefore, warned that only chi-square power computations involving proportions that average toward the middle values are to be relied upon heavily. Although there are no major differences in accuracy between the latter two algorithms, the Fleiss (1973) formula appears to be slightly more accurate at the larger sample sizes, so it is the basis for the interactive chi-square power analysis program.

To perform the power calculations, the program queries the user for the necessary information. By changing the values of input parameters within reasonable limits, the researcher can quickly put together a picture of the power characteristics of the planned research. The program first asks the user to select which of the following three aspects of power he wants computed: power, sample size, or detectable difference in proportions. The user is next asked for the alpha level to be used. If, for example, sample size is to be computed, the user is asked to enter the two population proportions. Last, the desired power is entered and the program displays the necessary sample size. Other options, if selected, are approached in a similar interactive manner.

Power is calculated using the following approximate formula from Fleiss (1973, p. 30, Formula 3.12):

$$\text{power} = \Pr\left\{ z > \frac{c_{\alpha/2}\sqrt{2PQ} - (P_2 - P_1)\sqrt{n}}{\sqrt{P_1Q_1 + P_2Q_2}} \right\},$$

where c is the critical value of the normal curve for alpha divided by 2, n is the common sample size per group, P_1 and P_2 are the population proportions, $Q = 1 - P$, P is the average of P_1 and P_2, and z is the standard normal deviate. This formula is coded in the program as FUNCTION PCHI. Normal curve critical values are calculated by FUNCTION CVZ, which calls FUNCTION ZPRB (described by Dunlap & Duffy, 1975) repeatedly to converge on the correct value within five-decimal accuracy. The probability of the standard normal deviate, z, is then computed by FUNCTION ZPRB. In order to calculate necessary sample size, the program uses FUNCTION PCHI repeatedly to get increasingly better estimates of the required sample size, converging on the integral value that will produce power equal to or greater than that requested. The calculation of group proportions detectable at a given power is also done iteratively, although the convergence is rapid.

Program Availability. The computational portions of the program are written in single-precision FORTRAN IV, and it runs on a DEC 2060 computer. Since the program is interactive, it is designed to run with a CRT terminal as the unit for input and output. Although pains were taken to avoid using DEC-specific FORTRAN statements, a few nonstandard features were included in the input statements to make the program run conveniently on the DEC system. First, the user will probably have to modify the statement assigning logical units if the program is to be run on computers other than DEC. Second, the symbol "$" in READ statements causes the cursor to remain at the end of a line on the CRT and, thus, is cosmetic but not necessary. Last, the asterisk in place of a FORMAT statement number allows free format input on the DEC system.

A listing of the program may be obtained at no cost from William P. Dunlap, Department of Psychology, Tulane University, New Orleans, Louisiana 70118.

REFERENCES

Camilli, G., & Hopkins, K. D. Applicability of chi-square to 2×2 contingency tables with small expected frequencies. *Psychological Bulletin*, 1978, **85**, 163-167.

Cohen, J. *Statistical power analysis for the behavioral sciences.* New York: Academic Press, 1977.

Dunlap, W. P. An interactive FORTRAN IV program for calculating power, sample size, or detectable differences in means. *Behavior Research Methods & Instrumentation*, 1981, **13**, 757-759.

Dunlap, W. P., & Duffy, J. A. FORTRAN IV functions for calculating exact probabilities associated with z, chi-square, t, and F values. *Behavior Research Methods & Instrumentation*, 1975, **7**, 59-60.

Fleiss, J. L. *Statistical methods for rates and proportions.* New York: Wiley, 1973.

Fry, T. C. The chi-square test of significance. *Journal of the American Statistical Association*, 1938, **33**, 513-525.

Pearson, E. S. The choice of statistical tests illustrated on the interpretation of data classed in a 2×2 table. *Biometrika*, 1947, **34**, 139-167.

Zelen, M., & Severo, N. C. Probability functions. In M. Abramowitz & I. A. Stegun (Eds.), *Handbook of mathematical functions.* New York: Dover, 1965.

Appendix

```
      OPEN (UNIT=5,DEVICE='TTY')
      OPEN (UNIT=6,DEVICE='TTY')
      WRITE(6,1)
1     FORMAT(' THIS PROGRAM HELPS YOU PLAN AN EXPERIMENT'/
      1 ' INVOLVING PROPORTIONS IN TERMS OF POWER — IN OTHER'/
      2 ' WORDS HOW LIKELY ARE YOU TO DETECT A SIGNIFICANT'/
      3 ' DIFFERENCE IN PROPORTIONS'/)
2     WRITE(6,3)
3     FORMAT(' DO YOU WISH TO COMPUTE:'/' 1 = POWER'/
      1' 2 = SAMPLE SIZE'/' 3 = DETECTABLE DIFFERENCE ',$)
      READ(5,*)K
      WRITE(6,4)
4     FORMAT(' WHAT IS YOUR ALPHA LEVEL ',$)
      READ(5,*)A
      GO TO (5,10,14),K
C
C  CALCULATE POWER
C
5     WRITE(6,6)
6     FORMAT(' ENTER P1,P2 ',$)
      READ(5,*)P1,P2
7     WRITE(6,8)
8     FORMAT(' ENTER N PER GROUP ( 0 FOR NEW PROBLEM ) ',$)
      READ(5,*)N
      IF (N.EQ.0) GOTO 2
      FN=N
      P=PCHI(N,P1,P2,A)
      WRITE(6,9)P
9     FORMAT(' YOUR POWER IS ',F8.4/)
      GOTO 7
C
C  CALCULATE SAMPLE SIZE
C
10    WRITE(6,6)
      READ(5,*)P1,P2
11    WRITE(6,12)
12    FORMAT(' ENTER DESIRED POWER ( 0 FOR NEW PROBLEM ) ',$)
      READ(5,*)P
      IF (P.EQ.0.0.OR.P.GE.1.0) GOTO 2
      PB=(P1+P2)/2.
      SPB=SQRT(2.*PB*(1.-PB))
      SP12=SQRT(P1*(1.-P1)+P2*(1.-P2))
      FN=((CVZ(A/2.)*SPB+CVZ(1.-P)*SP12)/(P1-P2))**2
      N=FN+.9
      WRITE(6,13)N
13    FORMAT(' N PER GROUP SHOULD BE',I5/)
      GOTO 11
C
C  CALCULATE DETECTABLE PROPORTIONS
C
14    WRITE(6,12)
      READ(5,*)P
      IF(P.EQ.0.0.OR.P.GE.1.0) GOTO 2
15    WRITE(6,8)
      READ(5,*)N
      IF (N.EQ.0) GOTO 2
16    WRITE(6,17)
17    FORMAT(' ENTER SMALLER PROP. (1 FOR NEW PROBLEM) ',$)
      READ(5,*)P1
      IF(P1.EQ.1.) GOTO 2
      BN=0.0
      IF(PCHI(N,P1,1.,A).LT.P) GOTO 21
      B1=.5
      B2=(P1+.5)/2.
      S1=PCHI(N,P1,B1,A)
      S2=PCHI(N,P1,B2,A)
18    BN=B2+(P-S2)*(B1-B2)/(S1-S2)
      IF(BN.LT.P1)BN=(P1+B2)/2.
      IF(BN.GT.1.)BN=(1.+B1)/2.
      SN=PCHI(N,P1,BN,A)
      IF(ABS(SN-P).LT..0001) GOTO 19
      B1=B2
      B2=BN
      S1=S2
      S2=SN
      GOTO 18
19    WRITE(6,20)BN
20    FORMAT(' THE UPPER PROPORTION IS ',F8.5/)
      GOTO 16
21    WRITE(6,22)
22    FORMAT(' NO SUCH PROPORTION EXISTS ')
      GOTO 16
      END
C
      FUNCTION CVZ(A)
C
C  COMPUTES ONE TAILED CRITICAL VALUES FROM THE NORMAL DISTRIBUTION
C     WITH AN ALPHA LEVEL = A
C
      F2=-.5
      F=.5
      PL2=ZPRB(-.5)
      PL2=ALOG(PL2)
      PE=ZPRB(.5)
      PLX=ALOG(A)
1     F1=F2
      F2=F
      PL1=PL2
      PL2=ALOG(PE)
      F=F1+(PLX-PL1)*(F2-F1)/(PL2-PL1)
      PE=ZPRB(F)
      IF(ABS(PE-A).GT..0000001) GOTO 1
      CVZ=F
      RETURN
      END
C
      FUNCTION ZPRB(Z)
C
C  COMPUTES 1-TAILED PROBABILITY OF STANDARD NORMAL DEVIATE Z
C  (INTEGRAL OF NORMAL CURVE FROM Z TO INFINITY)
C
      X=Z
      IF(Z.LT.0.0)X=-Z
      ZPRB=1.0
      IF(X.GT.5.612) GOTO 1
      ZX=.39894228*EXP((-X)*X/2.0)
      T=1.0/(1.0+.2316419*X)
      B=.31938153*T
      TT=T*T
      B=B-.356563782*TT
      TT=TT*T
      B=B+1.781477937*TT
      TT=TT*T
      B=B-1.821255978*TT
      TT=TT*T
      B=B+1.330274429*TT
      ZPRB=1.0-ZX*B
1     IF(Z.GT.0.0)ZPRB=1.0-ZPRB
      RETURN
      END
C
      FUNCTION PCHI(N,P1,P2,A)
C
C  COMPUTES POWER FOR A 2 X 2 CHI-SQUARE - FLEISS, 1973, P.30.
C     N = COMMON SAMPLE SIZE PER GROUP
C     P1 & P2 ARE POPULATION PROPORTIONS
C     A = ALPHA LEVEL
C
      FN=N
      D=ABS(P1-P2)
      PB=(P1+P2)/2.
      SPB=SQRT(2.*PB*(1.-PB))
      SP12=SQRT(P1*(1.-P1)+P2*(1.-P2))
      ZB=(SQRT(FN)*D-CVZ(A/2.)*SPB)/SP12
      PCHI=1.-ZPRB(ZB)
      RETURN
      END
```

(Accepted for publication May 7, 1982.)